
1. Постановка задачи 
 

Получить из файла статистику среднесуточной температуры за календарный год. 
Вывести на экран номера месяцев, в которых среднемесячная температура наименее отличается 
от среднегодовой. 
 
2. Уточнение исходных и выходных данных и ограничений на них 
 
 2.1. Исходные данные 

Во входном файле input.txt содержатся строки с информацией о средней 
температуре за каждый календарный день года в формате: DD.MM TTT.T, где DD и MM — день 
и месяц, а TTT.T — температура в указанный день. Например, 23.03 11.4. Количество строк 
заранее неизвестно. Количество дней в каждом месяце неизвестно (но не превышает количество 
дней в календарном месяце). Первые 6 символов каждой строки файла имеют строго заданный 
формат — 2 цифры, точка, 2 цифры, пробел. В оставшейся части строки содержится 
вещественное число, в котором дробная часть (если она есть) отделена точкой от целой части. 
Дробная часть может отсутствовать. Число может быть отрицательным (знак минус перед 
числом). Порядок указания дней в файле не определен, значит будем считать, что он 
произвольный. Так как первое число (DD) — номер дня месяца, считаем его целым числом от 1 
до 31. Так как второе число (MM) — номер месяца, считаем его целым числом от 1 до 12. 
Ограничения на температуру не заданы, считаем ее вещественным числом от –50 до +50, с 
одним десятичным знаком после "запятой". 
 
 2.2. Выходные данные 

Необходимо вывести на экран несколько чисел — номеров месяцев, среднемесячная 
температура которых наименее отличается от среднегодовой. Возможно, такое число будет 
только одно. Так как результат — номер месяца, считаем его целым числом от 1 до 12. 
 
3. Выбор метода решения 
 

3.1. Анализ исходных данных и выбор используемой структуры данных 
В задаче не требуется использовать информацию о конкретной суточной температуре 

какого-либо дня месяца. Анализируется лишь среднемесячная и среднегодовая температура. За 
один проход (одну линейную операцию чтения) по исходным данным можно, извлекая 
информацию о номере месяца из каждой входной строки, накапливать сумму месячных 
температур по каждому месяца, запоминая при этом количество обработанных дней каждого 
месяца. Для этого достаточно хранить для каждого месяца сумму температур и счетчик 
количества обработанных дней месяца. Для этого достаточно двух массивов из 12-ти элементов 
(по одному для каждого месяца) — из вещественных и из целых чисел. 

 
3.2. Выбор метода решения 
Считываем построчно данные из файла по каждому дню, извлекаем из них номер месяца 

и суточную температуру. Используем полученный номер месяца как индекс элементов 
массивов: в массиве, хранящем сумму температур за каждый месяц, увеличим сумму на 
величину прочитанной температуры, а в массиве количества дней месяца увеличим число на 
единицу. 

После чтения исходных данных приведем накопленные данные к более удобному для 
дальнейшей работы виду. Сложим все 12 элементов массива температур, а также все 12 

элементов массива дней месяца. Поделив после этого первую сумму на вторую, получим 
среднегодовую температуру. Так как после добавления суммы температур месяца к сумме 
годовой температуры, для каждого месяца желательно будет иметь уже не сумму температур за 
месяц, а среднемесячную температуру, поделим значение каждой ячейки на количество дней 
месяца и запишем на то же место. После этой операции в массиве температур будут храниться 
среднемесячные температуры за каждый месяц. 

Найдем среднемесячную температуру, наименее отличающуюся от среднегодовой. 
Найдем минимальное отклонение элементов массива температур от среднегодовой температуры 
(поиск минимума в массиве из 12-ти элементов). 

Сравним отклонение каждого элемента массива температур от полученного 
минимального отклонения. Для каждого элемента, у которого выполнится равенство, выведем 
его номер на экран. 
 
4. Составление алгоритма 
 
 4.1. Обобщенная блок-схема алгоритма 
 

 
 

Начало 

Чтение исходных 
данных. Сбор 
статистики 

Вычисление 
среднегодовой и 
среднемесячных 

температур 

Вычисление 
минимального 
отклонения от 
среднегодовой 
температуры 

Нахождение номеров 
месяцев, имеющих 

минимальное 
отклонение 

конец 



 4 

 4.2. Блок-схема алгоритма 
 

начало 

i – номер месяца 
r – температура 

Не конец 
файла? 

Читать i 
Читать r 
A[i]:=A[i]+r 
B[i]:=B[i]+1 Avg:=0 

K:=0 

i=1, 12 

A[i]:=0 
B[i]:=0 

Да 

Нет 

i=1,12 

Avg:=Avg+A[i] 
A[i]:=A[i]/B[i] 
K:=K+B[i] 

B[i] ≠0 

Нет 

Да 

A 

Avg:=Avg/K 
Min:=1000 

A 

i=1,12 

Avg:=Avg+A[i] 
A[i]:=A[i]/B[i] 
K:=K+B[i] 

B[i] ≠0 
|A[i]–Avg|<Min 

Нет 

Да 

конец 

i=1,12 

|A[i]–Avg|=Min 

Нет 

Да 

Вывод i 

5. Листинг программы 
 
var a:array [1..12] of real;      { Температура в каждом месяце } 
    b:array [1..12] of integer;   { Число дней в каждом месяце  } 
  k,                              { Количество дней в году      } 
  i:integer;                      { Вспомогательная переменная  } 
  f:text;                         { Исходный файл               } 
  c,                              { Вспомогательная переменная  } 
  m1,m2:char;                     { Цифры номера месяца         } 
  avg,    { Сумма температур в году и среднегодовая температура } 
  min,    { Наименьшее отклонение от среднегодовой температуры  } 
  r:real;                         { Температура дня             } 
begin 
  { Обнуляем суммы температур месяца и счетчики количества дней } 
  for i:=1 to 12 do 
  begin 
    a[i]:=0; 
    b[i]:=0 
  end; 
  { Открываем файл для чтения } 
  Assign(f,'input.txt'); 
  reset(f); 
  while not eof(f) do       { Пока нет конца файла } 
  begin 
    read(f,c,c,c,m1,m2);    { Читаем цифры номера месяца } 
    readln(f,r);            { Читаем температуру дня } 
    { Вычисляем номер месяца } 
    i:=(ord(m1)-ord('0'))*10+ord(m2)-ord('0');   
    a[i]:=a[i]+r;           { Добавляем температуру дня к сумме 
                              температур месяца } 
    inc(b[i])         { Увеличиваем счетчик количества дней месяца } 
  end; 
  Close(f);                 { Закрываем файл } 
  avg:=0;                   { Обнуляем сумму температур года       } 
  k:=0;                     { Обнуляем счетчик числа дней года     } 
  for i:=1 to 12 do         { Перебираем все 12 месяцев } 
    if b[i]<>0 then         { Если в месяце есть хоть один день, 
    begin                     температура которого известна        } 
      avg:=avg+a[i];        { Добавляем сумму температур месяца 
                              к сумме температур года              } 
      a[i]:=a[i]/b[i];      { Вычисляем среднемесячную температуру } 
      k:=k+b[i]             { Добавляем число дней месяца  
                              к числу дней года                    } 
    end; 
  if k<>0 then              { Если в году есть хоть один день      } 
    avg:=avg/k;             { Вычисляем среднегодовую температуру  } 
  min:=1000;                { Ищем минимальное отклонение          } 
  for i:=1 to 12 do 
    if (b[i]<>0) and (abs(a[i]-avg)<min) then {Для непустых месяцев} 
      min:=abs(a[i]-avg); 
  for i:=1 to 12 do         { Для каждого месяца } 
    if abs(a[i]-avg)=min then { Если его температура минимально  
                              отличается от среднегодовой          } 
      writeln(i);           { Выводим номер месяца на экран        } 
  readln 
end. 

6. Пример работы программы 
6.1. Исходные данные 

12.01 12.01 12.01 12.01 ----25252525    
30.09 1530.09 1530.09 1530.09 15    
14.14.14.14.02 02 02 02 ----8888    
26.02 26.02 26.02 26.02 ----12121212    
09.03 309.03 309.03 309.03 3    
18.03 10.518.03 10.518.03 10.518.03 10.5    
07.04 1207.04 1207.04 1207.04 12    
27.04 1827.04 1827.04 1827.04 18    
15.05 2215.05 2215.05 2215.05 22    
28.05 2028.05 2028.05 2028.05 20    
07.06 2507.06 2507.06 2507.06 25    
30.06 2730.06 2730.06 2730.06 27    
04.07 3004.07 3004.07 3004.07 30    
15.07 2515.07 2515.07 2515.07 25    
30.07 26.430.07 26.430.07 26.430.07 26.4    
12.08 2312.08 2312.08 2312.08 23    
31.08 22.331.08 22.331.08 22.331.08 22.3    
01.09 20.801.09 20.801.09 20.801.09 20.8    
17.09 2217.09 2217.09 2217.09 22    
14.10 1714.10 1714.10 1714.10 17    
19.10 1619.10 1619.10 1619.10 16    
01.11 1001.11 1001.11 1001.11 10    
17.11 517.11 517.11 517.11 5    
12.12 12.12 12.12 12.12 ----7777    
28.12 28.12 28.12 28.12 ----5.45.45.45.4    

6.2. Выходные данные 
4444    
    


